Dinamica dei Sistemi Aziendali

Prova parziale del 31 ottobre 2025

Esercizio 1 (21 punti). Una società di prodotti alimentari possiede tre stabilimenti, che possono produrre quattro diversi tipi di latte a lunga conservazione: intero, parzialmente scremato, senza lattosio intero e senza lattosio parzialmente scremato. Gli stabilimenti possono lavorare al massimo rispettivamente 400, 600 e 800 tonnellate (t) di latte a settimana. A causa dei diversi processi produttivi, a partire da una tonnellata di latte si ottengono 0.9 t di latte intero o parzialmente scremato e 0.8 t di latte senza lattosio (intero o parzialmente scremato). Si stima che il mercato richieda ogni settimana 500 t di latte intero, 400 t di latte parzialmente scremato, 50 t di latte senza lattosio intero e 30 t di latte senza lattosio parzialmente scremato. Il costo di lavorazione (in euro) di una tonnellata di latte dipende sia dallo stabilimento utilizzato che dal latte prodotto ed è indicato nella seguente tabella:

Stabilimento	Intero	Parz. scremato	Senza lattosio intero	Senza lattosio parz. scremato
1	6	5	8	7
2	7	6	9	8
3	8	7	10	9

Se uno stabilimento produce latte senza lattosio, è previsto un costo fisso aggiuntivo settimanale di 1000 euro. Per uno stabilimento passare dalla produzione di latte con lattosio alla produzione di latte senza lattosio (o viceversa) richiede un certo tempo di *setup* che è preferibile evitare. Ogni stabilimento può quindi produrre ogni settimana solo latte con lattosio oppure solo latte senza lattosio.

La società deve decidere ogni settimana quante tonnellate di latte saranno utilizzate da ogni stabilimento per la produzione dei quattro tipi di latte, rispettando i vincoli indicati sopra, con l'obiettivo di minimizzare il costo totale, dato dalla somma dei costi di lavorazione e dei costi fissi.

a) Scrivere un modello di Programmazione Lineare Intera per aiutare la società a risolvere il problema. Indicare le variabili decisionali, la funzione obiettivo ed i vincoli.

b) Scrivere nella sintassi di AMPL il modello di PLI formulato al punto a). Scrivere separatamente il file `.mod` ed il file `.dat`.

Esercizio 2 (10 punti). Considerare il problema di pianificare un progetto costituito da 6 attività che sono soggette a vincoli di precedenza. Il tempo di svolgimento (in giorni) ed i predecessori di ogni attività sono indicati nella seguente tabella:

Attività	A	B	C	D	E	F
Tempo di svolgimento	6	8	5	7	4	3
Predecessori	B	–	A,D	A,B	C,F	C,D

a) Rappresentare il problema di pianificare il progetto come un problema di cammino di costo massimo su un grafo opportuno.

b) Calcolare il minimo e massimo istante di inizio di ogni attività.

c) Calcolare la minima durata del progetto e trovare tutte le attività critiche.

Soluzioni

Esercizio 1.

a) Indichiamo gli stabilimenti con l'indice $i = 1, 2, 3$ ed i tipi di latte con l'indice j (intero $j = 1$, parzialmente scremato $j = 2$, senza lattosio intero $j = 3$, senza lattosio parzialmente scremato $j = 4$).

Definiamo le seguenti variabili:

x_{ij} = numero di tonnellate di latte utilizzate dallo stabilimento i per produrre il latte di tipo j,

$$y_i = \begin{cases} 1 & \text{se lo stabilimento } i \text{ produce almeno un tipo di latte senza lattosio,} \\ 0 & \text{altrimenti,} \end{cases}$$

$$z_i = \begin{cases} 1 & \text{se lo stabilimento } i \text{ produce almeno un tipo di latte con lattosio,} \\ 0 & \text{altrimenti,} \end{cases}$$

per ogni $i = 1, 2, 3$ e $j = 1, 2, 3, 4$.

Indicando con c_{ij} il costo di lavorazione di una tonnellata di latte per produrre il latte di tipo j nello stabilimento i, il problema si può formulare mediante il seguente modello di PLI:

$$\min \sum_{i=1}^{3} \sum_{j=1}^{4} c_{ij} x_{ij} + 1000 \sum_{i=1}^{3} y_i \tag{1}$$

$$x_{11} + x_{12} + x_{13} + x_{14} \leq 400 \tag{2}$$

$$x_{21} + x_{22} + x_{23} + x_{24} \leq 600 \tag{3}$$

$$x_{31} + x_{32} + x_{33} + x_{34} \leq 800 \tag{4}$$

$$0.9(x_{11} + x_{21} + x_{31}) \geq 500 \tag{5}$$

$$0.9(x_{12} + x_{22} + x_{32}) \geq 400 \tag{6}$$

$$0.8(x_{13} + x_{23} + x_{33}) \geq 50 \tag{7}$$

$$0.8(x_{14} + x_{24} + x_{34}) \geq 30 \tag{8}$$

$$x_{ij} \leq 800 z_i \qquad \forall\ i = 1, 2, 3,\ j = 1, 2 \tag{9}$$

$$x_{ij} \leq 800 y_i \qquad \forall\ i = 1, 2, 3,\ j = 3, 4 \tag{10}$$

$$y_i + z_i = 1 \qquad \forall\ i = 1, 2, 3 \tag{11}$$

$$x_{ij} \geq 0 \qquad \forall\ i = 1, 2, 3,\ j = 1, \ldots, 4$$

$$y_i, z_i \in \{0, 1\} \qquad \forall\ i = 1, 2, 3$$

Nella funzione obiettivo la prima somma rappresenta il costo di lavorazione, mentre la seconda i costi fissi dovuti alla produzione di latte senza lattosio.

I vincoli (2)–(4) garantiscono che la capacità produttiva di ogni stabilimento sia rispettata.

I vincoli (5)–(8) garantiscono che la domanda settimanale di latte sia soddisfatta.

I vincoli (9)–(10) legano tra loro le variabili x, y e z.

I vincoli (11) garantiscono che ogni stabilimento produce o solo latte con lattosio o solo latte senza lattosio.

b) Nella sintassi di AMPL il file `.mod` può essere scritto nel modo seguente:

```
#----------- parametri ---------------------

param n integer > 0;
```

```
# numero di tipi di latte

set TLconL within {1..n};
# insieme dei tipi di latte con lattosio

set TLsenzaL within {1..n};
# insieme dei di latte senza lattosio

param m integer > 0;
# numero di stabilimenti produttivi

param CostoLav{i in 1..m, j in 1..n} >= 0;
# CostoLav[i,j] = costo di lavorazione di una t di latte
# per produrre il latte di tipo j nello stabilimento i

param CostoFisso >= 0;
# costo fisso aggiuntivo settimanale
# se viene prodotto un tipo latte senza lattosio

param Capacita{i in 1..m} >= 0;
# Capacita[i] = numero max di t di latte che possono essere lavorate
# ogni settimana nello stabilimento i

param Domanda{j in 1..n} >= 0;
# Domanda[j] = numero di t di latte di tipo j richiesto ogni settimana dal mercato

param CoefconL > 0;
# da una t di latte si ottengono CoefconL t di latte con lattosio

param CoefsenzaL > 0;
# da una t di latte si ottengono CoefsenzaL t di latte senza lattosio

#------------ variabili ------------------

var x{i in 1..m, j in 1..n} >= 0;
# x[i,j] = numero di t di latte utilizzate
# dallo stabilimento i per produrre il latte di tipo j

var y{i in 1..m} binary;
# y[i] = 1 se lo stabilimento i produce latte senza lattosio,
#        0 altrimenti

var z{i in 1..m} binary;
# z[i] = 1 se lo stabilimento i produce latte con lattosio,
#        0 altrimenti

#------------ funzione obiettivo ------------------

minimize CostoTotale: sum{i in 1..m, j in 1..n} CostoLav[i,j]*x[i,j]
+ sum{i in 1..m} CostoFisso*y[i];

#------------ vincoli ------------------

s.t. VCapProd{i in 1..m}: sum{j in 1..n} x[i,j] <= Capacita[i];
```

```
s.t. VDom1{j in TLconL}: sum{i in 1..m} CoefconL*x[i,j] >= Domanda[j];

s.t. VDom2{j in TLsenzaL}: sum{i in 1..m} CoefsenzaL*x[i,j] >= Domanda[j];

s.t. VProdconL{i in 1..m, j in TLconL}: x[i,j] <= Capacita[i]*z[i];

s.t. VProdsenzaL{i in 1..m, j in TLsenzaL}: x[i,j] <= Capacita[i]*y[i];

s.t. Valt{i in 1..m}: y[i] + z[i] = 1;
```

mentre il corrispondente file `.dat` può essere scritto come segue:

```
param n := 4;

set TLconL := 1, 2;

set TLsenzaL := 3, 4;

param m := 3;

param CostoLav: 1 2 3 4 :=
1 6 5 8 7
2 7 6 9 8
3 8 7 10 9;

param CostoFisso := 1000;

param Capacita :=
1 400
2 600
3 800;

param Domanda :=
1 500
2 400
3 50
4 30;

param CoefconL := 0.9;

param CoefsenzaL := 0.8;
```

Esercizio 2.

a) Il problema di trovare la durata minima del progetto equivale a trovare un cammino di costo massimo dal nodo o al nodo d sul seguente grafo:



b) Per trovare il minimo e massimo istante di inizio di ogni attività, applichiamo il metodo CPM. Un ordinamento topologico dei nodi è il seguente:



Il minimo e massimo istante di inizio di ogni attività sono riportati nella tabella seguente:

Attività	Minimo istante inizio	Massimo istante inizio
1	0	0
2	0	0
3	8	8
4	14	14
5	21	21
6	26	26
7	29	29
8	33	33

c) La minima durata del progetto è 33 giorni e tutte le 6 attività sono critiche.